

Description  
Options  
Also see

Quick start  
Remarks and examples

Menu  
Methods and formulas

Syntax  
References

## Description

`sts list` lists the estimated survivor, failure, or the Nelson–Aalen estimated cumulative (integrated) hazard function. See [ST] `sts` for an introduction to this command.

`sts list` can be used with single- or multiple-record or single- or multiple-failure `st` data.

## Quick start

Kaplan–Meier survivor function using `stset` data

```
sts list
```

Show the number at risk and the survivor function only at specified times

```
sts list, risktable(10 20 30 40 50)
```

Survivor function for each level of `v1`

```
sts list, by(v1)
```

Results of above saved in `mydata.dta`

```
sts list, by(v1) saving(mydata)
```

Show numbers at risk and survivor functions for groups of `v1` only at specified times

```
sts list, by(v1) risktable(10(10)50)
```

Same as above, but report survivor functions for each group side by side

```
sts list, by(v1) risktable(10(10)50) compare
```

Failure function

```
sts list, failure
```

Failure function adjusted for  $v2 = 0$

```
sts list, failure adjustfor(v2)
```

Same as above, but adjusted for mean value of `v2`

```
sts list, failure adjustfor(v2, atmeans)
```

Failure function adjusted for  $v2 = 10$  and stratified on levels of `svar`

```
sts list, failure adjustfor(v2, at(v2=10)) strata(svar)
```

## Menu

Statistics > Survival analysis > Summary statistics, tests, and tables > List the survivor or related function

## Syntax

```
sts list [if] [in] [, options]
```

<i>options</i>	Description
<b>Main</b>	
<u>survival</u>	report Kaplan–Meier survivor function; the default
<u>failure</u>	report Kaplan–Meier failure function
<u>cumhaz</u>	report Nelson–Aalen cumulative hazard function
by( <i>varlist</i> )	estimate separate functions for each group formed by <i>varlist</i>
strata( <i>varlist</i> )	stratify on different groups of <i>varlist</i> ; requires <code>adjustfor()</code>
<u>adjustfor</u> ( <i>varlist</i> [, <i>suboptions</i> ])	adjust the estimates to specific values of <i>varlist</i> ; default is zero values
<b>Options</b>	
<u>level</u> (#)	set confidence level; default is <code>level(95)</code>
<u>risktable</u>	report estimated survivor or related function at all distinct observed times; the default
<u>risktable</u> ( <i>numlist</i> )	report estimated survivor or related function at specified times
<u>tmax</u> (#)	display estimated function for $t \leq \#$
<u>tmin</u> (#)	display estimated function for $t \geq \#$
<u>enter</u>	report number who entered and number lost as pure censored instead of censored minus lost (net lost)
<u>noshow</u>	do not show st setting information
<u>compare</u>	report groups of estimated functions side by side; requires <code>by()</code> or <code>strata()</code>
<u>saving</u> ( <i>filename</i> [, <code>replace</code> ])	save results to <i>filename</i> ; use <code>replace</code> to overwrite existing <i>filename</i>
<u>atrisk0</u>	report number who entered as number at risk at time 0 instead of a zero

You must `stset` your data before using `sts list`; see [\[ST\] stset](#).

`fweights`, `iweights`, and `pweights` may be specified using `stset`; see [\[ST\] stset](#).

`atrisk0` does not appear in the dialog box.

## Options

### Main

`survival`, `failure`, and `cumhaz` specify the function to report.

`survival` specifies that the Kaplan–Meier survivor function be listed. This option is the default if a function is not specified.

`failure` specifies that the Kaplan–Meier failure function  $1 - S(t + 0)$  be listed.

`cumhaz` specifies that the Nelson–Aalen estimate of the cumulative hazard function be listed.

`by`(*varlist*) estimates a separate function for each by-group. By-groups are identified by equal values of the variables in *varlist*. `by()` may not be combined with `strata()`.

`strata`(*varlist*) requests estimates of the survivor or failure function stratified on variables in *varlist*. It requires specifying `adjustfor()` and may not be combined with `by()` or `cumhaz`.

`adjustfor(varlist [, suboptions])` adjusts the estimate of the survivor or failure function to specific values of *varlist*. The default is to adjust to 0 values, that is, to produce a baseline function. If you want to adjust the function to values different from 0, you can use `adjustfor()`'s `at()` suboption.

*suboptions* are `atzeros` (the default), `atmeans`, `atomeans`, `atbase`, and `at()`; see [ST] [adjust-for\\_option](#).

Say you want to produce the survivor function adjusted to age of patients and the ages in your sample are 40–60 years. If you want to adjust the function to age 40, type

```
. sts list, adjustfor(age, at(age=40))
```

Or if you want to adjust the function to the mean age, type

```
. sts list, adjustfor(age, atmeans)
```

`adjustfor()` is not available with `cumhaz`.

If you specify `adjustfor()` with `by()`, `sts list` fits separate Cox regression models for each group, using the `adjustfor()` variables as covariates. The separately calculated baseline functions are then retrieved.

If you specify `adjustfor()` with `strata()`, `sts list` fits a stratified-on-group Cox regression model, using the `adjustfor()` variables as covariates. The stratified baseline function is then retrieved.

If adjustment to covariate values other than 0 is requested, the function is estimated at the specified covariate values.

#### Options

`level(#)` specifies the confidence level, as a percentage, for the Greenwood pointwise confidence interval of the survivor or failure function or for the pointwise confidence interval of the Nelson–Aalen cumulative hazard function; see [U] [20.8 Specifying the width of confidence intervals](#).

`risktable` or `risktable(numlist)` reports the estimated survivor, failure, or cumulative hazard function.

`risktable`, the default, reports the results at all the distinct observed times or, if functions are being compared, at about 10 times chosen over the observed interval. At time 0, the number at risk is zero. At other times, the number at risk records the number of subjects who are under observation and did not fail immediately prior to the reported time.

`risktable(numlist)` reports the results at time points specified in *numlist*. You can think of `risktable(numlist)` as a table representation of `sts graph, risktable(numlist)`, except the number of failures is always reported with `sts list`, and the number at risk at time 0 is not the number of subjects in a study but is a zero. To match `sts graph`, you can use the `atrisk0` option to report the number at risk at time 0 as the number who entered at that time instead of a zero.

`risktable(5 10 20 30 50 90)` would display the function at the designated times.

`risktable(10 20 to 100)` would display the function at times 10, 20, 30, 40, ..., 100.

`risktable(0 5 10 to 100 200)` would display the function at times 0, 5, 10, 15, ..., 100, and 200.

`risktable(20)` would display the function at time 20.

See [example 1](#) under *Remarks and examples*.

`tmax(#)` specifies that the estimated function and other summaries be reported only for  $t \leq \#$ . This option does not affect the calculation of the function, rather the portion that is displayed, but it does affect the calculation of certain summaries such as the number of failures and the number lost. `tmax(#)` is ignored when `#` is smaller than the maximum time specified in `risktable(numlist)`.

`tmin(#)` specifies that the estimated function and other summaries be reported only for  $t \geq \#$ . This option does not affect the calculation of the function, rather the portion that is displayed, but it does affect the calculation of certain summaries such as the number who entered. `tmin(#)` is ignored when `#` is larger than the minimum time specified in `risktable(numlist)`.

`enter` specifies that the table contain the number who entered and, correspondingly, that the number lost be displayed as the pure number censored rather than censored minus entered (net lost). The logic underlying this is explained in [Counting the number lost to censoring](#) in [ST] `sts`.

`noshow` prevents `sts list` from showing the key `st` variables. This option is seldom used because most people type `stset`, `show` or `stset, noshow` to set whether they want to see these variables mentioned at the top of the output of every `st` command; see [ST] `stset`.

`compare` is specified only with `by()` or `strata()`. It compares the survivor, failure, or cumulative hazard functions and lists them side by side rather than first one and then the next.

`saving(filename[, replace])` saves the results in a Stata data file (`.dta` file).

`replace` specifies that `filename` be overwritten if it exists.

The following option is available with `sts list` but is not shown in the dialog box:

`atrisk0` specifies to report the number who entered as the number at risk at time 0 instead of the default zero value. By definition, the number at risk at time 0 is a zero. `atrisk0` is useful with `risktable(numlist)` to match the at-risk table as reported by the corresponding `sts graph`, `risktable(numlist)` specification.

## Remarks and examples

`sts list` reports the estimated survivor, failure, or cumulative hazard function and, in the absence of probability weights, its standard errors and confidence intervals.

The Kaplan–Meier product-limit estimate of the survivor function is displayed by default or if you specify the `survival` option. Use the `failure` option to display the Kaplan–Meier product-limit estimate of the failure function. Or use the `cumhaz` option to display the Nelson–Aalen estimate of the cumulative hazard function.

Only three of `sts list`’s options—`adjustfor()`, `by()`, and `strata()`—modify the calculation of a function. All the other options merely determine how the results of the calculation are displayed.

Specify the `by()` option to perform the calculation separately on the different groups and the `compare` option to see the results side by side.

Specify the `adjustfor()` option to calculate an adjusted survivor or failure function. If you also specify `by()` or `strata()`, this further modifies how the adjustment is made. With `by()`, the function is estimated after fitting separate Cox regression models for each group using the `adjustfor()` variables as covariates. With `strata()`, which requires `adjustfor()`, the function is estimated after fitting a stratified-on-group Cox regression model using the `adjustfor()` variables as covariates. Only the estimate of the function is reported with `adjustfor()`.

By default, `sts list` displays the estimated function and other summaries such as the number at risk, the number of failures, and the number lost (censored) at all distinct observed times. You can specify the `risktable(numlist)` option to see the results at specific times. This is recommended for large datasets. In the presence of delayed entry or gaps, the command reports the number lost as net lost—the number of censored minus the number who entered. You can specify the `enter` option to see the number lost and the number who entered separately; see [Counting the number lost to censoring](#) in [ST] `sts`.

You can also restrict the maximum and minimum ranges of the observed times using the `tmax(#)` and `tmin(#)` options. These options do not affect the calculation of the estimated function but will affect the computation of certain summaries such as the number of failures and the number lost and will affect the display of the results; see [example 1](#).

We demonstrate many of `sts list`'s features in [ST] `sts`. This discussion picks up where that entry leaves off.

### ► Example 1: Using option `risktable()`

By default, `sts list` will bury you in output. With the Stanford heart transplant data introduced in [ST] `stset`, the following commands produce 154 lines of output.

```
. use https://www.stata-press.com/data/r19/stan3
(Heart transplant data)
```

```
. stset, noshow
```

```
. sts list, by(posttran)
```

Kaplan-Meier survivor function

By variable: `posttran`

Time	At risk	Fail	Net lost	Survivor function	Std. error	[95% conf. int.]	
posttran=0							
1	103	1	3	0.9903	0.0097	0.9331	0.9986
2	99	3	3	0.9603	0.0195	0.8976	0.9849
3	93	3	3	0.9293	0.0258	0.8574	0.9657
(output omitted)							
1400	1	0	1	0.2359	0.1217	0.0545	0.4882
posttran=1							
1	0	0	-3	1.0000	.	.	.
2	3	0	-3	1.0000	.	.	.
(output omitted)							
5.1	14	1	0	0.9286	0.0688	0.5908	0.9896
6	13	0	-1	0.9286	0.0688	0.5908	0.9896
(output omitted)							
1799	1	0	1	0.1420	0.0546	0.0566	0.2653

Note: Net lost equals the number lost minus the number who entered.

`risktable()` is the solution. Here is another detailed, but more useful, view of the heart transplant data:

```
. sts list, risktable(10 40 to 170) by(posttran)
```

Kaplan-Meier survivor function

By variable: posttran

Time	At risk	Fail	Net lost	Survivor function	Std. error	[95% conf. int.]	
posttran=0							
10	74	9	34	0.8724	0.0346	0.7858	0.9256
40	31	4	11	0.6781	0.0601	0.5446	0.7801
70	16	1	5	0.6126	0.0704	0.4603	0.7339
100	10	1	0	0.5616	0.0810	0.3900	0.7022
130	9	1	1	0.5054	0.0903	0.3199	0.6646
160	7	2	5	0.4422	0.0986	0.2480	0.6204
posttran=1							
10	16	6	-31	0.9286	0.0688	0.5908	0.9896
40	41	9	-11	0.7391	0.0900	0.5140	0.8716
70	43	8	-5	0.6002	0.0841	0.4172	0.7423
100	40	2	1	0.4814	0.0762	0.3271	0.6198
130	37	1	0	0.4687	0.0752	0.3174	0.6063
160	36	18	18	0.4561	0.0742	0.3076	0.5928


Notes: Survival-time summaries shown for specified times in (0, 1799].

Net lost equals the number lost minus the number who entered.

We specified `risktable(10 40 to 170)`, but the time 170 was not displayed because the numlist 10 40 to 170 ends at 160. Specifying the numlist as 10 40 to 160 would have been clearer.

We obtain the same numbers at risk as produced by `sts graph`:

```
. sts graph, risktable(10 40 to 170, failevents lastfailure) by(posttran)
> xlabel(10 40 to 160 170) tmax(170)
```



The number of failures for the last interval reported by `sts graph` is 0 in the first group and 1 in the second group, whereas `sts list` reported 2 and 18, respectively. We specified `tmax(170)` with `sts graph` to restrict the range of the observed times to the maximum of 170, whereas `sts list` used the observed maximum times for each group (1,400 and 1,799). If we want to match the numbers of failures, we could specify `tmax(170)` with `sts list`:

```
. sts list, risktable(10 40 to 170) by(posttran) tmax(170)
```

Kaplan-Meier survivor function

By variable: posttran

Time	At risk	Fail	Net lost	Survivor function	Std. error	[95% conf. int.]	
posttran=0							
10	74	9	34	0.8724	0.0346	0.7858	0.9256
40	31	4	11	0.6781	0.0601	0.5446	0.7801
70	16	1	5	0.6126	0.0704	0.4603	0.7339
100	10	1	0	0.5616	0.0810	0.3900	0.7022
130	9	1	1	0.5054	0.0903	0.3199	0.6646
160	7	0	1	0.4422	0.0986	0.2480	0.6204
posttran=1							
10	16	6	-31	0.9286	0.0688	0.5908	0.9896
40	41	9	-11	0.7391	0.0900	0.5140	0.8716
70	43	8	-5	0.6002	0.0841	0.4172	0.7423
100	40	2	1	0.4814	0.0762	0.3271	0.6198
130	37	1	0	0.4687	0.0752	0.3174	0.6063
160	36	1	-1	0.4561	0.0742	0.3076	0.5928

Notes: Survival-time summaries shown for specified times in (0, 170].

Net lost equals the number lost minus the number who entered.

◀

## ► Example 2: Nelson–Aalen estimated cumulative hazard

Continuing with [example 1](#), we can produce similar output for the Nelson–Aalen estimated cumulative hazard by specifying the cumhaz option:

```
. sts list, cumhaz risktable(10 40 to 170) by(posttran)
```

Nelson-Aalen cumulative hazard function

By variable: posttran

	At		Net	Cumulative	Std.		
Time	risk	Fail	lost	hazard	error	[95% conf. int.]	
posttran=0							
10	74	9	34	0.1349	0.0391	0.0764	0.2382
40	31	4	11	0.3824	0.0871	0.2448	0.5976
70	16	1	5	0.4813	0.1124	0.3044	0.7608
100	10	1	0	0.5646	0.1400	0.3473	0.9178
130	9	1	1	0.6646	0.1720	0.4002	1.1037
160	7	2	5	0.7896	0.2126	0.4658	1.3385
posttran=1							
10	16	6	-31	0.0714	0.0714	0.0101	0.5071
40	41	9	-11	0.2929	0.1176	0.1334	0.6433
70	43	8	-5	0.4981	0.1360	0.2916	0.8507
100	40	2	1	0.7155	0.1542	0.4691	1.0915
130	37	1	0	0.7418	0.1564	0.4908	1.1214
160	36	18	18	0.7689	0.1587	0.5130	1.1523

Notes: Survival-time summaries shown for specified times in (0, 1799].

Net lost equals the number lost minus the number who entered.

We can use the `compare` option to compare the cumulative hazard functions of the two groups side by side:

```
. sts list, cumhaz risktable(10 40 to 170) by(posttran) compare
Nelson-Aalen cumulative hazard function
By variable: posttran
```

posttran		0	1
Time	10	0.1349	0.0714
	40	0.3824	0.2929
	70	0.4813	0.4981
	100	0.5646	0.7155
	130	0.6646	0.7418
	160	0.7896	0.7689

◀

## Video example

How to calculate the [Kaplan-Meier survivor](#) and [Nelson-Aalen cumulative hazard functions](#)

## Methods and formulas

See [\[ST\] sts](#).

## References

See [\[ST\] sts](#) for references.

## Also see

[\[ST\] sts](#) — Generate, graph, list, and test the survivor and related functions

[\[ST\] sts generate](#) — Create variables containing survivor and related functions

[\[ST\] sts graph](#) — Graph the survivor or related function

[\[ST\] sts test](#) — Test equality of survivor functions

[\[ST\] stset](#) — Declare data to be survival-time data

[\[ST\] adjustfor\\_option](#) — Adjust survivor and related functions for covariates at specific values

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).